LEARNING MACROECONOMIC THEORY AND POLICY ANALYSIS VIA MICROCOMPUTER SIMULATION

W. F. Edwards, Murray State University

ABSTRACT

This paper presents a representative set of three macroeconomic models out of a larger set of models used to teach macroeconomic theory and policy analysis at the intermediate level. Besides solving the models by hand, students solve them on the microcomputers. This pedagogical approach is designed to accomplish three objectives: (1) to answer student criticisms that most macroeconomic textbooks do not present them with concrete, numerical models, (2) to improve students' mathematical skills, and (3) to reinforce computer skills.

INTRODUCTION

A few years ago when I began teaching macroeconomics, I soon realized that a deficiency common to almost all textbooks was that they tended to utilize economic concepts which were highly abstract for most students and they never seemed to reduce the concepts to a concrete, operational level, or model which the students could easily manipulate. Student complaints about a variety of texts that we tried through the semesters first made me aware of this problem. Finally, in response to their continued pleas for more concrete examples that they "could get a handle on," I began developing a series of macroeconomic models of the economy to illustrate the major schools of thought. At first we used the mainframe computer at the university but later when we acquired a number of microcomputers which were readily available to students, we modified our models to run on the micros. Over the years, the series of examples, along with their attendant explanations, has gradually grown into the main pedagogical device that we use in teaching Intermediate Macroeconomic Theory. The purpose of this paper is to present three of the models that are used in class, to discuss briefly the advantages and disadvantages of this technique, and to inform you of future plans for the project.

The Classical School of Thought

Two models of the classical system are presented. The first contains only production and consumption sectors--no government sector. Illustrative simulations, using this model, include:

1. Impact of a policy decision by the Fed to change the money supply.
2. Effect of a shift in the economy's production function caused by a change in technology.
3. Impact of a change in the profit expectations of businessmen manifesting itself through a shift in the investment function. Figure 1 presents the BASIC computer program to accomplish these simulations. Data entry is explained in program statements 140-220. Figure 2 presents the mathematical model, the values of parameters, exogenous variables, and policy variables, data statements for each simulation, and the output for each simulation. The first simulation demonstrates the classical conclusion that increasing the money supply merely influences the

FIGURE 1
BASIC PROGRAM TO SOLVE THE FIRST CLASSICAL MODEL

```
10 REM: MACRO1
20 REM: THE BEGINNING CLASSICAL MODEL
30 REM: STRUCTURE OF THE MODEL IS:
40 REM: Y=F1+F2*L+F3*L^2                    ECONOMY'S PRODUCTION FUNCTION
50 REM: LD=F2/(2*F3)-(1/(2*F3))*W           LABOR DEMAND
60 REM: LS=G1+G2*W                          LABOR SUPPLY
70 REM: LD=LS                               EQUIL. IN LABOR MKT.
80 REM: MD=K*P*Y                            NOMINAL MONEY DEMAND
90 REM: MS=MO                               NOMINAL MONEY SUPPLY
100 REM: MD=MS                              EQUIL. IN MONEY MKT.
110 REM: SS=S0+S1*R                         SUPPLY OF LOANABLE FUNDS
120 REM: I=I0-I1*R                          DEMAND FOR LOANABLE FUNDS
130 REM: SD=SS                              EQUIL. IN MKT. FOR LOANABLE FUNDS
140 REM: YOU MUST PROVIDE PARAMETERS AND EXOGENOUS VALUES
150 REM: IN DATA STATEMENTS 600-650, AS FOLLOWS:
160 REM: STATEMENT # ; ITEMS TO PROVIDE
170 REM:   600           PARAMETERS OF PRODUCTION FUNC, F1,F2,F3
180 REM:   610           PARAMETERS OF LABOR SUPPLY FUNCTION,G1,G2
190 REM:   620           PARAMETER OF NOMINAL DEMAND FOR MONEY,K
200 REM:   630           EXOGENOUS MONEY SUPPLY,MO
210 REM:   640           PARAMETERS OF SUPPLY FUNC FOR LOAN. FUNDS,S0,S1
220 REM:   650           PARAMETERS OF DEM. FOR LOAN. FUNDS,I0,I1
230 READ F1,F2,F3:READ G1,G2:READ K:READ MO:
240 READ S0,S1:READ I0,I1
250 W=(F2-2*F3*G1)/((2*F3*G2)+1)
260 L=(G1+G2*F2)/((2*F3*G2)+1)
270 Y=F1+F2*L-F3*L^2
280 P=MO/(K*Y)
290 R=(I0-S0)/(S1+I1)
300 SS=S0+S1*R
310 I=I0-I1*R
320 PRINT:PRINT:PRINT "MODEL SOLUTION"
330 PRINT "REAL Y = ";Y
340 PRINT "NOMINAL Y = ";Y*P
350 PRINT "QUANTITY OF LABOR=";L
360 PRINT "REAL WAGE=";W
370 PRINT "NOMINAL WAGE=";W*P
380 PRINT "PRICE LEVEL=";P
390 PRINT "REAL INTEREST RATE=";R
410 PRINT "SAVINGS=";SS
420 PRINT "INVESTMENT=";I
600 DATA 5,80,1
610 DATA 5,.5
620 DATA .5
630 DATA 700
640 DATA -10,800
650 DATA 120,500
700 END
```

FIGURE 2
THE CLASSICAL MODEL: GOVERNMENT SECTOR EXCLUDED

Conditions Simulated: (1) Impact of a change in the money supply.
(2) Effect of a shift in the economy's production function caused by a change in technology.
(3) Impact of a change in profit expectations of businessmen manifesting itself through a shift in the investment function.

Mathematical Model

[1] $Y = F1 + (F2)L - (F3)L^2 = 5 + 80L - L^2$ Economy's production function.
[2] $LD = \frac{F2}{2F3} - \frac{1}{2F3} W = 40 - .5W$ Economy's demand function for labor; derived from the economy's production function.
[3] $LS = G1 + (G2)W = 5 + .5W$ Economy's supply function for labor.
[4] $LD = LS$ Equilibrium condition for the labor market.
[5] $MD = (K)(P)(Y) = (.5)(P)(Y)$ Economy's demand function for money.
[6] $MS = MO = 700$ Economy's supply of money: Exogenously determined by the Fed.
[7] $MD = MS$ Equilibrium condition for the money market.
[8] $SD = I0 - I1(R) = 120 - 500R$ Economy's demand function for loanable funds.
[9] $SS = S0 + S1(R) = -10 + 800R$ Economy's supply function for loanable funds.
[10] $SD = SS$ Equilibrium condition for the market for loanable funds.

Where:

L = Quantity of labor; LD = Quantity of labor demanded; LS = Quantity of labor supplied.
MD = Quantity of money demanded; MS = Quantity of money supplied.
SD = Quantity of loanable funds demanded; SS = Quantity of loanable funds supplied.
W = The real wage; P = The price level; Y = Real output; R = Rate of interest.

Basic Solution
Data Statements:

600 DATA 5,80,1	Parameters for the economy's production function.
610 DATA 5,.5	Parameters for economy's supply function for labor.
620 DATA .5	Parameter, K, for the economy's demand for money.
630 DATA 700	Exogenous money supply.
640 DATA -10,800	Parameters of the economy's supply function for loanable funds.
650 DATA 120,500	Parameters of the economy's demand function for loanable funds.

Solution:

Y = 1298.75	L = 22.5	Nominal Wage = 37.73	R = .10
Nominal Y = 1400	W = 35	P = 1.078	Equilibrium SS = SD = 70

Simulation 1: An Increase in the Money Supply from 700 to 1400.
Changes in Data Statements:
630 DATA 1400
Solution:

Y = 1298.75	L = 22.5	Nominal Wage = 75.46	R = .10
Nominal Y = 2800	W = 35	P = 2.16	Equilibrium SS = SD = 70

Simulation 2: Shift in the Economy's Production Function.
Changes in Data Statements:

630 DATA 700	Return money supply to 700.
600 DATA 80,80,.8	Change parameters of production function.

Solution:

Y = 1580	L = 25	Nominal Wage = 35.44	R = .10
Nominal Y = 1400	W = 40	P = .87	Equilibrium SS = SD = 70

Simulation 3: Increased Profit Expectations Raise Demand for Loanable Funds.
Changes in Data Statements:

600 DATA 5,80,1	Return parameters of production function.
650 DATA 150,500	Change parameters of demand for loanable funds.

Solution:

Y = 1298.75	L = 22.5	Nominal Wage = 37.73	R = .12
Nominal Y = 1400	W = 35	P = 1.078	Equilibrium SS=SD = 88.46

price level--when the money supply is doubled, the price level doubles, and so on. Equilibrium values of real variables are left unchanged. The second simulation shows how an increase in the economy's production function, through perhaps improved technology, can increase real output and the real wage as well as lower the price level. The third simulation demonstrates that increased profit expectations under the classical model, manifesting themselves through an increased demand for loanable funds, will tend to increase the interest rate as well as increase the equilibrium values of the quantity of loanable funds demanded and supplied.

The second classical model introduces the government sector into the model equations, making adjustments in variable definitions where necessary, and simulates the following situations: 1. A growing government with a balanced budget. 2. Growth in government financed by borrowing. Figure 3 presents the computer program to accomplish these simulations. Data entry is explained in program statements 210-320. Figure 4 then presents the mathematical model, the values of parameters, exogenous variables, and policy variables, as well as the output of each simulation. The first simulation demonstrates the classical conclusion that if government finances its own growth through taxation, the result will be a partial "crowding out" of the consuming sector. The second simulation shows that if the expansion is financed through borrowing rather than through taxation, both the consuming sector and the investment sector (i.e., the private demand for loanable funds) will be partially "crowded out."

The Keynesian School of Thought

In class, we first develop the familiar two-sector model consisting of a consumption sector and a production sector. We solve the algebraic model under the assumption that investment is exogenous. Second, we introduce the government sector with its attendant modifications. Third, we return to the question of investment and introduce an investment function that changes the investment variable from an exogenous to an endogenous status. Finally, we add the monetary sector, which enables us to develop the familiar IS/LM analysis. Since the Keynesian models, at this stage, are all linear, and since our students have been exposed to matrix algebra by the time they take this course, in the junior or senior year, we use a matrix format to set up the models for computer solution:

$$[A]\,[X] = [B]$$

$$[X] = [A]^{-1}\,[B]$$

where [A] is the matrix of coefficients, [X] the vector of solution variables, and [B] the right hand side vector. The computer program to solve this model is presented in Figure 5. Data entry begins with statement 700 showing the number of equations in the system. Beginning in statement 710, the elements of the [A] matrix are entered and the last element of each statement is the corresponding element of the [B] matrix. This will be clear when examining Figure 6. In Figure 6, we present the final Keynesian Model, which includes the government and monetary sectors. This model specification makes it very easy for students to enter new values of policy variables such as the money supply, government spending, or the tax rate, and explore the consequences of such changes on the endogenous variables. It is also very easy to play

FIGURE 3
BASIC PROGRAM TO SOLVE THE SECOND CLASSICAL MODEL

```
10 REM: MACRO2
20 REM: EXTENSIONS OF THE BEGINNING CLASSICAL MODEL
30 REM: STRUCTURE OF THE MODEL IS:
40 REM: Y=F1+F2*L-F3*L^2                    ECONOMY'S PRODUCTION FUNCTION
50 REM: LD=F2/(2*F3)-(1/(2*F3))*W           LABOR DEMAND
60 REM: LS=G1+G2*W                          LABOR SUPPLY
70 REM: LD=LS                               EQUIL. IN LABOR MKT.
80 REM: MD=K*P*Y                            NOMINAL MONEY DEMAND
90 REM: MS=MO                               NOMINAL MONEY SUPPLY
100 REM: MD=MS                              EQUIL. IN MONEY MKT.
110 REM: G=GO                               GOVERNMENT SPENDING
120 REM: TR=TRO                             TRANSFER PAYMENTS
130 REM: TX=TXO                             TAXES
140 REM: SS=S0+S1*R                         SUPPLY OF LOANABLE FUNDS
150 REM: I=I0-I1*R                          PRIVATE DEMAND FOR LOANABLE FUNDS
160 REM: SD=I+G+TR-TX                       TOTAL DEMAND FOR LOANABLE FUNDS
170 REM: SD=SS                              EQUIL. IN MKT. FOR LOANABLE FUNDS
180 REM: YD=Y-TX+TR                         DISPOSABLE INCOME
190 REM: C=YD-SS                            CONSUMPTION EXPENDITURES
200 REM: Y=C+I+G                            AGGR. SUPPLY=AGGRE. DEMAND
210 REM: YOU MUST PROVIDE PARAMETERS AND EXOGENOUS VALUES
220 REM: IN DATA STATEMENTS 600-680, AS FOLLOWS:
230 REM: STATEMENT # ; ITEMS TO PROVIDE
240 REM:    600          PARAMETERS OF PRODUCTION FUNC. F1,F2,F3
250 REM:    610          PARAMETERS OF LABOR SUPPLY FUNCTION,G1,G2
260 REM:    620          PARAMETER OF NOMINAL DEMAND FOR MONEY,K
270 REM:    630          EXOGENOUS MONEY SUPPLY,MO
280 REM:    640          EXOGENOUS GOVERNMENT SPENDING, GO
290 REM:    650          EXOGENOUS TRANSFER PAYMENTS, TRO
300 REM:    660          EXOGENOUS TAXES,TXO
310 REM:    670          PARAMETERS OF SUPPLY FUNC FOR LOAN. FUNDS,S0,S1
320 REM:    680          PARAMETERS OF PRIV. DEM. FOR LOAN. FUNDS,I0,I1
330 READ F1,F2,F3:READ G1,G2:READ K:READ MO:READ GO
340 READ TRO:READ TXO:READ S0,S1:READ I0,I1
350 W=(F2-2*F3*G1)/((2*F3*G2)+1)
360 L=(G1+G2*F2)/((2*F3*G2)+1)
370 Y=F1+F2*L-F3*L^2
380 P=MO/(K*Y)
390 R=(I0+GO+TRO-TXO-S0)/(S1+I1)
400 SS=S0+S1*R
410 I=I0-I1*R
420 YD=Y-TXO+TRO
430 C=YD-SS
440 PRINT:PRINT:PRINT "MODEL SOLUTION"
450 PRINT "REAL Y = ";Y
460 PRINT "NOMINAL Y = ";Y*P
470 PRINT "QUANTITY OF LABOR=";L
480 PRINT "REAL WAGE=";W
490 PRINT "NOMINAL WAGE=";W*P
500 PRINT "PRICE LEVEL=";P
510 PRINT "REAL INTEREST RATE=";R
520 PRINT "SAVINGS=";SS
530 PRINT "INVESTMENT=";I
540 PRINT "CONSUMPTION=";C
600 DATA 5,80,1
610 DATA 5,.5
620 DATA .5
630 DATA 700
640 DATA 0
650 DATA 0
660 DATA 0
670 DATA -10,800
680 DATA 120,500
700 END
```

FIGURE 4
THE CLASSICAL MODEL: GOVERNMENT SECTOR INCLUDED

Conditions Simulated: (1) A growing government with a balanced budget.
(2) A growing government financed by borrowing.

Mathematical Model

```
[1]  Y = 5 + 80L - L^2         Economy's production function--same as Fig. 2.
[2]  LD = 40 - .5W             Economy's demand function for labor.
[3]  LS = 5 + .5W              Economy's supply function for labor.
[4]  LD = LS                   Equilibrium condition for the labor market.
[5]  MD = (.5)(P)(Y)           Economy's demand function for money.
[6]  MS = 700                  Economy's supply of money: Exogenously
                                 determined by the Fed.
[7]  MD = MS                   Equilibrium condition for the money market.
[8]  G = Go                    Government spending--an exogenous variable.
[9]  TR = TRo                  Government transfer payments--exogenous.
[10] TX = TXo                  Government taxes--exogenous.
[11] SS = -10 + 800R           Economy's supply function for loanable funds.
[12] I = 120 - 500R            Private sector demand for loanable funds.
[13] SD = I + (G + TR) - TX    Economy's total demand function for loanable
                                 funds (includes government demand).
[14] SD = SS                   Equilibrium condition for the market for
                                 loanable funds.
```

Basic Solution

Data Statements:

```
600 DATA 5,80,1
610 DATA 5,.5
620 DATA .5
630 DATA 700
640 DATA 0  ┐
650 DATA 0  ├  G + TR - TX = 0
660 DATA 0  ┘
670 DATA -10,800
680 DATA 120, 500
```

Solution:

```
Y = 1298.75          C = 1228.75    SS = 70    L = 22.5    Nominal Wage = 37.73
Nominal Y = 1400     I = 70         SD = 70    W = 35      R = .1
```

Simulation 1: Government Spending and Transfer Payments Rise to 40, Taxes Rise to 40 to Maintain Balanced Budget.

Changes in Data Statements:

```
640 DATA 40
660 DATA 40
```

Solution:

```
Y = 1298.75          C = 1188.75    SS = 70    L = 22.5    Nominal Wage = 37.73
Nominal Y = 1400     I = 70         SD = 70    W = 35      R = .1
```

Simulation 2: Government Spending and Transfer Payments = 40; Taxes = 0. Government Runs a Deficit.

Changes in Data Statements:

```
660 DATA 0
```

Solution:

```
Y = 1298.75          C = 1204.14    SS = 94.62    L = 22.5    Nominal Wage = 37.73
Nominal Y = 1400     I = 54.62      SD = 94.62    W = 35      R = .13
```

FIGURE 5
SIMULTANEOUS EQUATIONS PROGRAM TO SOLVE KEYNESIAN MODELS

```
100 REM:  MACRO3
130 REM: Any BASIC, any CRT.
150 CLEAR:CLS:DEFINT J,K,L,M,N
180 PRINT
190 READ N
200 IF N>0 THEN 230
210 PRINT:BEEP
220 PRINT "There must be at least 1 !":GOTO 180
230 DIM A(N,N),R(N),V(N)
240 PRINT
270 FOR J=1 TO N
280 FOR K=1 TO N
290 READ A(J,K)
300 NEXT
320 READ R(J)
330 NEXT
335 GOSUB 390
340 PRINT"The solution is"
350 PRINT:FOR J=1 TO N
360 PRINT V(J)
370 NEXT:END
380 PRINT:PRINT STRINGS$(40,205):PRINT:RETURN
390 IF N>1 THEN 410
400 V(1)=R(1)/A(1,1):RETURN
410 FOR K=1 TO N-1:M=K+1
420 L=K
430 Q=ABS(A(M,K))-ABS(A(L,K))
440 IF Q>0 THEN L=M
450 IF M<N THEN M=M+1:GOTO 430
460 IF L=K THEN 490
470 FOR J=K TO N:SWAP A(K,J),A(L,J):NEXT
480 SWAP R(K),R(L)
490 M=K+1
500 Q=A(M,K)/A(K,K):A(M,K)=0
510 FOR J=K+1 TO N
520 A(M,J)=A(M,J)-Q*A(K,J):NEXT
530 R(M)=R(M)-Q*R(K)
540 IF M<N THEN M=M+1:GOTO 500
550 NEXT
560 V(N)=R(N)/A(N,N)
570 FOR M=N-1 TO 1 STEP -1
580 Q=0:FOR J=M+1 TO N:Q=Q+A(M,J)*V(J)
590 V(M)=(R(M)-Q)/A(M,M):NEXT:NEXT:RETURN
700 DATA 10
710 DATA 0,-.9,1,0,0,0,0,0,0,0,5
720 DATA -1,1,0,1,-1,0,0,0,0,0,0
730 DATA -.3,0,0,1,0,0,0,0,0,0,0
740 DATA 0,0,0,0,1,0,0,0,0,0,100
750 DATA -.01,0,0,0,0,1,4500,0,0,0,1600
760 DATA 0,0,0,0,0,0,0,0,0,1,800
770 DATA 1,0,-1,0,0,-1,0,0,0,-1,0
780 DATA -.2,0,0,0,0,0,5000,1,0,0,800
790 DATA 0,0,0,0,0,0,0,0,1,0,1000
800 DATA 0,0,0,0,0,0,0,-1,1,0,0,
810 END
900 END
```

FIGURE 6
KEYNESIAN MODEL: INCLUDING GOVERNMENT AND MONETARY SECTORS.

Mathematical Model

[1] C = Ca + bYd	Consumption is a linear function of disposable income.
[2] Yd = Y − TX + TR	Disposable income is equal to total income--taxes + transfer payments.
[3] TX = tY	Taxes are a proportion, t, of total income.
[4] TR = TRa	Transfer payments are exogenous with a value of TRa.
[5] I = Ia − eR + fY	Investment expenditures are inversely related to the interest rate and directly related to Y.
[6] G = Ga	Government spending is exogenous with a value of Ga.
[7] Y = C + I + G	Total spending (income) is the sum of C, I, and G. Equilibrium condition for the product market.
[8] MD = M1 + M2(Y) − M3(R)	Economy's demand for money is directly related to Y and inversely related to R.
[9] MS = Ma	The money supply is exogenous with a value of Ma.
[10] MS = MD	Equilibrium condition for the money market.

Matrix Specification.

$$\begin{bmatrix} 0 & -b & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & 1 & 0 & 1 & -1 & 0 & 0 & 0 & 0 & 0 \\ -t & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ -f & 0 & 0 & 0 & 0 & 1 & e & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & -1 & 0 & 0 & -1 & 0 & 0 & 0 & -1 \\ -M2 & 0 & 0 & 0 & 0 & 0 & M3 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 1 & 0 \end{bmatrix} \begin{bmatrix} Y \\ Yd \\ C \\ TX \\ TR \\ I \\ R \\ MD \\ MS \\ G \end{bmatrix} = \begin{bmatrix} Ca \\ 0 \\ 0 \\ TRa \\ Ia \\ Ga \\ 0 \\ M1 \\ Ma \\ 0 \end{bmatrix}$$

Values of Parameters and Exogenous Variables.

Ca = 5	e = 4500	M2 = .2	Ga = 800
b = .9	f = .01	M3 = 5000	Ia = 1600
t = .3	M1 = 800	TRa = 100	Ma 1000

Data Statements.

```
700 DATA 10
710 DATA 0,-.9,1,0,0,0,0,0,0,0,5
720 DATA -1,1,0,1,-1,0,0,0,0,0,0
730 DATA -.3,0,0,1,0,0,0,0,0,0,0
740 DATA 0,0,0,0,1,0,0,0,0,0,100
750 DATA -.01,0,0,0,0,1,4500,0,0,0,1600
760 DATA 0,0,0,0,0,0,0,0,0,1,800
770 DATA 1,0,-1,0,0,-1,0,0,0,-1,0
780 DATA -.2,0,0,0,0,0,5000,1,0,0,800
790 DATA 0,0,0,0,0,0,0,0,1,0,1000
800 DATA 0,0,0,0,0,0,0,-1,1,0,0
```

Model Solution.

Y = 4953,704
Yd = 3567,593
C = 3215,833
TX = 1486,111
TR = 100
I = 937,870
R = .1561
MD = 1000
MS = 1000
G = 800

the "what-if game" with behavioral equations such as the consumption function, the investment function, the demand for money, and so on to see how shifts in the parameters of these functions affect the model solution. Multiplier concepts are also easy to illustrate by, for example, changing autonomous investment and noting the impact on *Y*.

<u>Advantages and Disadvantages of the Computer Modeling Approach to Learning Macroeconomics</u>

We have found this method of instruction to have three advantages. First, students seem to develop a fuller and deeper understanding of the economic theory when they work with concrete models such as these. They are better able to constructively criticize the models, understand their assumptions and suggest ways of making the models more realistic. Second, students, of necessity, must reinforce their mathematical skills to successfully manipulate the smaller models by hand, which we have them do before working on the computer. Third, it gives the students another opportunity to reinforce their computer skills, which we believe should be done more frequently in classes outside the computer discipline itself.

Unfortunately, these advantages do not come without a cost. The main cost we have discovered is that it is very difficult to cover as much material as is usually covered in a more traditional macroeconomics class. For example, we have not yet been successful in getting beyond the Keynesian model to the Rational Expectations Model. We are currently trying to develop a series of models to help students explore the rational expectations philosophy, but we suspect that to cover this material might require a two semester sequence. Another possibility is to offer the full course at the graduate level rather than at the junior/senior level as it is currently offered. If this were done, it is likely that the graduate students could cover the entire range of models in one semester. We intend to explore this alternative in the future.